Fausto Fabian Crespo Fernández

Resolución EDO con series usando MAXIMA

En Maxima se puede declarar ecuaciones diferenciales al igual que se declaran las variables y se usa ' antes del diff para evitar que Maxima calcule la derivada. Po ejemplo para la ecuación:

$$(2x+3)y' + y(x-1) = 0$$

Usamos

edo:(2*x+3)*'diff(y,x)+y*(x-1)=0;

que al ejecutar shift+return da

(%o5) $(2\,x+3)\left(\frac{d}{d\,x}\,y\right)+(x-1)\,y=0$

Que se puede resolver por la variables separables con

ode2(edo,y,x);

(%o6) $y=\%c\,\%e^{1.25\log(2\,x+3)-0.5\,x}$

El software Maxima no tiene la opción "series" al igual que en Macsyma, pero podemos usar desarrollo de Taylor la parte izquierda de la EDO y las condiciones iniciales para hallar la solución de series de potencias de la EDO, Por ejemplo, si queremos resolver:

$$y'' - xy = 0 \;\; con\; y(0) = 1\; y\; y'(0) = 1$$

Podemos usar para asignar los valores iniciales:

atvalue(y(x),x=0,1);atvalue('diff(y(x),x),x=0,1);

y para el desarrollo de Taylor de la parte izquierda de la EDO:

edoTaylorserie:taylor('diff(y(x),x,2)-x*y(x),x,0,3);

(%o9)/T/ $\left.\frac{d^2}{d\,x^2}\,y(x)\right|_{x=0}+\left(\left.\frac{d^3}{d\,x^3}\,y(x)\right|_{x=0}-1\right)x+\frac{\left(\left.\frac{d^4}{d\,x^4}\,y(x)\right|_{x=0}-2\right)x^2}{2}-\frac{\left(3\left(\left.\frac{d^2}{d\,x^2}\,y(x)\right|_{x=0}\right)-\left.\frac{d^5}{d\,x^5}\,y(x)\right|_{x=0}\right)x^3}{6}$

Ahora esta serie es idénticamente a 0(lado derecho de la EDO que es 0) y por tanto podemos igualar cada coeficiente a 0 y resolver:

solve([at('diff(y(x),x,2),x=0)=0,(at('diff(y(x),x,3),x=0)-1)=0,(at('diff(y(x),x,4),x=0)

-2)/2=0,-((3*(at('diff(y(x),x,2),x=0))-at('diff(y(x),x,5),x=0)))/6=0],

[at('diff(y(x),x,2),x=0),at('diff(y(x),x,3),x=0),at('diff(y(x),x,4),x=0),at('diff(y(x),x,5),x=0)]);

Lo que da

$$(\%o10)\quad [[\left.\frac{d^2}{d x^2} y(x)\right|_{x=0} = 0, \left.\frac{d^3}{d x^3} y(x)\right|_{x=0} = 1, \left.\frac{d^4}{d x^4} y(x)\right|_{x=0} = 2, \left.\frac{d^5}{d x^5} y(x)\right|_{x=0} = 0]]$$

Con lo que tenemos las derivadas de y(x) evaluadas en 0 y podemos saber el desarrollo de y(x) en serie alrededor de 0:

$$y(t) = y(0) + y'(0)t + \frac{y''(0)}{2}t^2 + \frac{y'''(0)}{6}t^3 + \frac{y^{(IV)}(0)}{24}t^4 + \frac{y^{(V)}(0)}{120}t^5 + O(t^6)$$

$$y(t) = 1 + t + \frac{1}{6}t^3 + \frac{1}{12}t^4 + O(t^6)$$

Otra vía de obtener la solución de una EDO mediante serie es la siguiente que se ilustrara con el ejemplo: supongamos que queremos hallar la solución de $y'' + xy' - y = 0$ mediante series de potencias:

y:sum(a[n]*x^n, n, 0, inf);

$$(\%o12)\quad \sum_{n=0}^{\infty} a_n x^n$$

Luego derivamos una vez y dos veces para obtener los otros términos de la EDO:

dy2:diff(y,x,2);

$$(\%o13)\quad \sum_{n=0}^{\infty} (n-1)\, n\, a_n x^{n-2}$$

dy1:diff(y,x);

$$(\%o14)\quad \sum_{n=0}^{\infty} n\, a_n\, x^{n-1}$$

Ahora debemos cambiar el índice n parar que empiece en 2 en dy2 y en 1 en dy1:

s2:subst(2,0,dy2);

$$(\%o15)\quad \sum_{n=2}^{\infty} (n-1)\, n\, a_n\, x^{n-2}$$

s1:subst(1,0,dy1);

$$(\%o16)\quad \sum_{n=1}^{\infty} n\, a_n\, x^{n-1}$$

Ahora introducimos factor x en s1 y la constante -1 en y :

s3:intosum(-y);

$$(\%o17)\quad \sum_{n=0}^{\infty} -a_n\, x^{n}$$

s4:intosum(x*s1);

$$(\%o18)\quad \sum_{n=1}^{\infty} n\, a_n\, x^{n}$$

Corremos los dices en s3 y s4 o sea $n = k - 2$:

s4:changevar(s4,n+2-k,k,n);

(%o19) $$\sum_{k=3}^{\infty} (k-2)\, a_{k-2}\, x^{k-2}$$

s3:changevar(s3,n+2-k,k,n);

(%o20) $$-\sum_{k=2}^{\infty} a_{k-2}\, x^{k-2}$$

Como se ve hay índices k(en s4 y s3) y n(en s1) por eso cambiamos todos los índices de k a n en s3 y s4:

niceindicespref:[n];

(%o21) $[n]$

niceindices(s4);

(%o22) $$\sum_{n=3}^{\infty} (n-2)\, a_{n-2}\, x^{n-2}$$

niceindices(s3);

(%o23) $$-\sum_{n=2}^{\infty} a_{n-2}\, x^{n-2}$$

Introducimos el -1 de s3 dentro de la sumatoria:

s3:intosum(s3);

(%o24) $$\sum_{k=2}^{\infty} -a_{k-2}\, x^{k-2}$$

Sumamos los 3 términos:

sumcontract(s2+s3+s4);

$$(\%o26) \quad \left(\sum_{n=3}^{\infty} (n-1)\, n\, a_n\, x^{n-2} + (n-2)\, a_{n-2}\, x^{n-2} - a_{n-2}\, x^{n-2}\right) + 2\, a_2 - a_0$$

Todos los coeficientes deben ser 0 o sea $a_2 = \frac{1}{2a_0}$ y para n>=3 se cumple la recurrencia:

rec:(n-1)*n*a[n]+(n-2)*a[n-2]-a[n-2];

$$(\%o27) \quad (n-1)\, n\, a_n + (n-2)\, a_{n-2} - a_{n-2}$$

solve(rec=0,a[n]);

$$(\%o28) \quad [a_n = -\frac{(n-3)\, a_{n-2}}{n^2 - n}]$$

Que se puede factorizar y % se refiere a la última expresión:

factor(%);

$$(\%o29) \quad [a_n = -\frac{(n-3)\, a_{n-2}}{(n-1)\, n}]$$

Definiendo ahora a0 y a1 (el símbolo $ al final significa que la instrucción continúa hasta ;)

a[0]:a0$

a[1]:a1$

a[n]:=-(n-3)/(n*(n-1))*a[n-2];

$$(\%o32) \quad a_n := \frac{-(n-3)}{n\, (n-1)}\, a_{n-2}$$

Ahora podemos construir la cantidad de términos de la serie solución que queramos, por ejemplo 7 términos:

sum(a[n]*x^n,n,0,7);

$$(\%o33) \quad \frac{a0\, x^6}{240} - \frac{a0\, x^4}{24} + \frac{a0\, x^2}{2} + a1\, x + a0$$